





Zadání domácí práce z předmětu UPS

Téma úlohy:

	V jazyce symbolických adres procesoru 80x86 vypracujte program pro řízení přípravku semafor.





































































vypracoval :		 David Ryšánek

číslo skupiny :	 240

datum :		� DATUM  \* VČETNĚFORMÁTU �11.05.97�

	Pro vytvoření programu pro semafor jsem si vymyslel fiktivní křižovatku. Ze západu na východ je hlavní silnice (nebo silnice s větším provozem) a silnice vedoucí ze severu na jih je vedlejší. Silnice vedoucí z jihu má pruh pro odbočení vpravo. Tento jízdní pruh využívá signál pro odbočení doprava.

	Semafory pro automobily a chodce se mohou rozsvěcet ve třech různých režimech, které lze přepínat přepínači na přípravku. Přepínač je podle tohoto programu sepnutý, pokud je v horní poloze. Pokud se přepínače nacházejí v jiném stavu než je potřeba pro nastavení určitého cyklu, tak je semafor přepnut na noční provoz tzn. bliká na všech semaforech pro automobily žlutá. Po nastavení správné kombinace přepínačů se začne provádět určitý cyklus až do doby, kdy se nastaví přepínači jiný cyklus nebo se ukončí program stiskem libovolné klávesy. Právě aktivní cyklus indikují LED vlevo na přípravku.



Nastavení přepínačů pro jednotlivé cykly:

Př1�Př2�Př3�Př4�Akce��on�off�off�off�Noční provoz��on�on�off�off�Cykl 1��on�on�on�off�Cykl 2��on�on�on�on�Cykl 3��

Popis cyklů:

	Cykl 1: Hlavní silnice je upřednostněna před vedlejší silnicí tzn. zelená v tomto směru svítí déle (asi dvakrát). Na jihu je signál pro odbočení doprava. Rozsvítí se chvíli po naskočení červené ve směru sever-jih a zhasne také dříve, než se rozsvítí žlutá. Semafory pro chodce mají zelenou jen ve „stabilním“ stavu křižovatky tj. jen když mají automobily v jednom směru zelenou a v druhém červenou nebo naopak. Při přepínání stavů semaforu mají všichni chodci červenou.

	Cykl 2: Je stejný jako Cykl 1 s tím rozdílem, že je dvakrát pomalejší.

	Cykl 3: Je stejný jako Cykl 1. Není zde však použit signál pro odbočení na silnici vedoucí z jihu.

Popis programu:

	Po spuštění programu se na monitoru vypíše zpráva o zavedení programu a program  rovnou začne provádět podle nastavení přepínačů jednotlivé cykly. Program lze ukončit stiskem jakékoliv klávesy. Vypíše se zpráva o konci programu a program skončí.

	Celý program jsem se snažil udělat strukturovaně. Používám procedury a makra pro snadnější orientaci v programu. Další možné cykly lze jednoduše dodefinovat pomocí volání maker a dodefinováním dekódovací tabulky přepínačů Tab v sekci DATASEG. 

	Tabulky Auta, Chodci a Ledky slouží k nastavení dat přípravku. Jejich změnou lze nastavit jednotlivé stavy semaforů pro auta, chodce nebo LED.

	Sekce CODESEG začíná nastavením kódového a datového segmentu atd. direktivou STARUPCODE. Následuje uvedení přípravku do počátečního stavu tj. zhasnutí všech semaforů a LED. Dále začíná hlavní program resp. cykl, který se vykonává dokud neukončíme program klávesou. Nejprve se testuje stav přepínačů a na základě rovnosti dat z tabulky a procedury Test_pr se zvolí příslušný cykl (voláním procedury, která se skládá z  maker, která postupně vyšlou data z tabulek do přípravku i s příslušným čekáním mezi stavy). Na konci cyklu se testuje stav klávesnice, a pokud nebyla stištěna klávesa, tak se cykl opakuje. V opačném případě se zhasnou všechny semafory a LED, vypíše se zpráva o konci programu a direktiva EXITCODE provede ukončení programu přerušením DOSu 21h.

Výpis programu Semafor:

;	 David Rysanek 1997  Program pro ovladani pripravku semafor	   

; registry:

;	    bx - data, ktera se maji vyslat vystupnim portem

;	    di - citac cyklu a pocet taktu pro cekani (1 takt = 55 ms);

IDEAL

MODEL SMALL

STACK 100h



LPT_port    EQU 378h

; definice maker

MACRO	A_Nastav index_pole,pocet_taktu      ;makro vysle data pro auta a pocka

	mov   bx,[Auta+2*index_pole]	 ;do bx data, ktera se maji vyslat

	call  Vysli_A			 ;vyslani dat portem

	mov   di,pocet_taktu		 ;nastaveni doby zpozdeni

	call  Delay			 ;cekaci procedura

ENDM

MACRO	Ch_Nastav index_pole,pocet_taktu ;makro vysle data pro chodce a pocka

	mov   bl,[Chodci+index_pole]

	call  Vysli_Ch

	mov   di,pocet_taktu

	call  Delay

ENDM

MACRO	L_Nastav index_pole,pocet_taktu  ;makro vysle data pro ledky a pocka

	mov   bl,[Ledky+index_pole]

	call  Vysli_L

	mov   di,pocet_taktu

	call  Delay

ENDM

MACRO	Print co			           ;makro vypise zadany text

	mov   dx,offset co

	mov   ah,09h

	int   21h

ENDM

; konec definice maker

DATASEG

Text1	   db	'Program na ovladani pripravku semafor.$'

Text2	   db	13,10,'Konec programu.$'

;

;C - cervena, Z - zluta, z - zelena, O - odboceni

;ctverice bitu : zapad, vychod, sever, jih (plati pro auta i chodce)

;		CZzOCZzOCZzOCZzO

Auta	   dw	0000000000000000b ;0   nic nesviti

	   dw	0100010001000100b ;1   sviti jen zlute

	   dw	0010001010001000b ;2   SJ cervena ZV zelena

	   dw	0100010010001000b ;3   SJ cervena ZV zluta

	   dw	1000100010001000b ;4   SJ cervena ZV cervena

	   dw	0010001011001100b ;5   SJ cervena+zluta ZV cervena

	   dw	1000100000100010b ;6   SJ zelena  ZV cervena

	   dw	1000100001000100b ;7   SJ zluta   ZV cervena

	   dw	1100110010001000b ;8   SJ cervena ZV cervena+zluta

	   dw	0010001010001001b ;9   SJ cervena ZV zelena J odboceni

;		CZCZCZCZ

Chodci	   db	00000000b	  ;0	nic nesviti

	   db	01011010b	  ;1	ZapVych=zelena, SevJih=cervena

	   db	10100101b	  ;2	ZapVych=cervena, SevJih=zelena

	   db	10101010b	  ;3	vsude cervena

;		nnnn1234

Ledky	   db	00000000b		;nic nesviti

	   db	00001000b		;sviti dioda c.1

	   db	00000100b		;sviti dioda c.2

	   db	00000010b		;sviti dioda c.3

	   db	00000001b		;sviti dioda c.4

;		x1234xxx      log. 0 - prepinac sepnut

Tab	   db	00111111b		;prepinac 1 sepnut

	   db	00011111b		;prepinac 2 sepnut

	   db	00001111b		;prepinac 3 sepnut

	   db	00000111b		;prepinac 4 sepnut

;

CODESEG

STARTUPCODE

	  mov ah,01h

	  int 16h			;vymaze ENTER z bufferu klavesnice

	  Print Text1		;uvodni text

	  A_Nastav   0,0		;zhasne auta

	  Ch_Nastav  0,0		;zhasne chodce

	  L_Nastav   0,0		;zhasne ledky

Znovu:	  call Test_pr		;stav tlacitek se vraci v al

	  cmp al,[Tab]		;porovna prepinace s tabulkou Vyp

	  jz C0 			;a zvoli prislusny ovladaci cykl.

	  cmp al,[Tab+1]		;Pokud je nastavena spatna kombinace

	  jz C1 			;prepinacu, tak nocni provoz

	  cmp al,[Tab+2]		;

	  jz C2 			;		       |

	  cmp al,[Tab+3]		;		       |

	  jz C3 			; <--------------------------------'

C0:	  L_Nastav   1,0

	  A_Nastav  1,24		;rozsviti zlutou vsude

	  A_Nastav  0,15		;zhasne zlutou (nocni provoz)

	  Jmp Stisk

C1:	  call Cykl1

	  jmp Stisk

C2:	  call Cykl2

	  jmp Stisk

C3:	  call Cykl3

Stisk:	  mov ah,01h		;testuje stisk klavesy

	  int 16h

	  jnz Konec		;konci, pokud byla stistena klavesa

	  jmp Znovu		;jinak se cyklus opakuje

Konec:	  A_Nastav   0,0		;zhasne auta

	  Ch_Nastav  0,0		;zhasne chodce

	  L_Nastav   0,0		;zhasne ledky

	  Print Text2		;text na konec programu

	  mov di,20

	  call Delay

EXITCODE

; definice procedur

; procedura, ktera vysle data na port 387H, zdroj dat je bx, nastavuji se bity D0 (data) a D1 (hodiny) portu 378H

PROC	  Vysli_A

	  mov  dx,LPT_port	       ;nastavi vystupni port

	  mov  cx,10h		       ;pocet cyklu (bitu na vyslani)

	  not  bx		        	       ;pouziva se nagativni logika

Nav1:	  xor  ax,ax		       ;nulovani ax

	  ror  bx,1		       ;rotace bx vpravo (v CF je vysil. bit)

	  adc  al,ah		       ;CF se ulozi do al

	  out  dx,al		       ;al resp. CF se vysle na vystupni port

	  or   al,00000010b	       ;nastavi se bit D1 na log.1

	  out  dx,al		       ;al na port - hodiny registru

	  loop Nav1		       ;skok, pokud cx<>0

	  ret

ENDP

; procedura, ktera vysle data na port 387H, zdroj dat je bl, nastavuji se bity D2 (data) a D3 (hodiny) portu 378H

PROC	  Vysli_Ch

	  mov  dx,LPT_port	       ;nastavi vystupni port

	  mov  cx,08h		       ;pocet cyklu (bitu na vyslani)

	  not  bl		       	       ;pouziva se nagativni logika

Nav2:	  xor  ax,ax		       ;nulovani ax

	  ror  bl,1		       	       ;rotace bl vpravo (v CF je vysil. bit)

	  adc  al,ah		       ;CF se ulozi do al

	  shl  al,1		       ;nastaveni dat na D2

	  shl  al,1

	  out  dx,al		       ;al resp. CF se vysle na vystupni port

	  or   al,00001000b	       ;nastavi se bit D3 na log.1

	  out  dx,al		       ;al na port - hodiny registru

	  loop Nav2		       ;skok, pokud cx<>0

	  ret

ENDP

; procedura, ktera vysle data na port 387H, zdroj dat je bl, nastavuji se bity D4 (data) a D5 (hodiny) portu 378H

PROC	  Vysli_L

	  mov  dx,LPT_port	       ;nastavi vystupni port

	  mov  cx,04h		       ;pocet cyklu (bitu na vyslani)

	  not  bl		      	       ;pouziva se nagativni logika

Nav3:	  xor  ax,ax		       ;nulovani ax

	  ror  bl,1		       	       ;rotace bl vpravo (v CF je vysil. bit)

	  adc  al,ah		       ;CF se ulozi do al

	  shl  al,1		       ;nastaveni dat na D4

	  shl  al,1		       ;	             |

	  shl  al,1		       ;	             |

	  shl  al,1		       ;  <---------------'

	  out  dx,al		       ;al resp. CF se vysle na vystupni port

	  or   al,00100000b	       ;nastavi se bit D5 na log.1

	  out  dx,al		       ;al na port - hodiny registru

	  loop Nav3		       ;skok, pokud cx<>0

	  ret

ENDP

; procedura, ktera ceka definovanou dobu (danou v registru di)

PROC	  Delay

	  cmp  di,0		       ;skoci, pokud je nastaveno 0 taktu

	  jz   Kon

	  mov  ah,0		       ;nastaveni cisla sluzby

	  int  1ah		                      ;poctu taktu od posledniho resetu

	  add  di,dx		       ;zjisti konecny pocet taktu

Nav4:	  mov  ah,0		       ;zjistuje aktualni pocet taktu

	  int  1ah		                      ;a porovnava ho s konecnym poctem

	  cmp  di,dx		       ;taktu danych v registru di

	  jnz  Nav4		       ;skace, dokud se oba pocty taktu

Kon:	  ret			       ;nerovnaji

ENDP

; procedura testujici stav prepinacu (stav vraci v al) 0XXXX111

PROC	  Test_pr

	  mov dx,LPT_port	       ;cislo vystupniho portu

	  mov ax,01h		       ;mastaveni pro multiplex

	  out dx,ax		       ;vyslat na port 378h 01h

	  inc dx			       ;nastavi se na 379h

	  in al,dx		                      ;precte stav prepinacu

	  ret

ENDP

; procedura Cykl1  ( 3.3s zelena ZV a 2.5s zelena SJ + odboceni vpravo)

PROC	  Cykl1

	  L_Nastav   2,0		;2. Led

	  A_Nastav   2,5		;SJ cervena ZV zelena

	  Ch_Nastav  2,10		;SJ zelena  ZV cervena

	  A_Nastav   9,60		;SJ cervena ZV zelena + J odboceni

	  A_Nastav   2,10		;SJ cervena ZV zelena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   5,8		;SJ cervena+zluta ZV cervena

	  A_Nastav   6,5		;SJ zelena  ZV cervena

	  Ch_Nastav  1,45		;SJ cervena ZV zelena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   7,10		;SJ zluta   ZV cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   8,8		;SJ cervena ZV cervena+zluta

	  ret

ENDP

; procedura Cykl2 (2x pomaleji nez v Cykl1 - bez odboceni)

PROC	  Cykl2

	  L_Nastav   3,0		;3. Led

	  A_Nastav   2,5		;SJ cervena ZV zelena

	  Ch_Nastav  2,150	;SJ zelena  ZV cervena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   5,8		;SJ cervena+zluta ZV cervena

	  A_Nastav   6,5		;SJ zelena  ZV cervena

	  Ch_Nastav  1,90		;SJ cervena ZV zelena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   7,10		;SJ zluta   ZV cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   8,8		;SJ cervena ZV cervena+zluta

	  ret

ENDP

; procedura Cykl3 (stejne rychle jako v cykl1, ale bez odboceni)

PROC	  Cykl3

	  L_Nastav   4,0		;4. Led

	  A_Nastav   2,5		;SJ cervena ZV zelena

	  Ch_Nastav  2,80		;SJ zelena  ZV cervena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   5,8		;SJ cervena+zluta ZV cervena

	  A_Nastav   6,5		;SJ zelena  ZV cervena

	  Ch_Nastav  1,45		;SJ cervena ZV zelena

	  Ch_Nastav  3,0		;vsude cervena

	  A_Nastav   7,10		;SJ zluta   ZV cervena

	  A_Nastav   4,5		;SJ cervena ZV cervena

	  A_Nastav   8,8		;SJ cervena ZV cervena+zluta

	  ret

ENDP

; konec definice procedur

END

;			      Konec programu

�STRÁNKA  �





�STRÁNKA  �5�










